Use NotInList Event to Add a Record to Combo Box

A combo box is used to select a value from a list.  If the desired value is not in the list, and one attempts to type it in, they may be greeted with the following error message:

Microsoft Knowledge Base Article 197526 includes two methods that allow one to respond to the Not in List event procedure.  This procedure can be used to add new values to the combo box.  These examples use the Orders form in the Northwind sample database.
Method 1: Using Code to Add a Record to a Table

( A new record is added to the Customers table, using information from an input box.
Method 2: Using a Form to Add a New Record

( A new record is added to the Customers table, by first opening the Customers form.  This allows one to add additional customer data (address, city, state, phone, etc.) to the record at the same time without presenting an endless series of input box prompts.

However, this method has a serious flaw—it will choke on a company name that includes an apostrophe (even though the record will ultimately be added to the Customer's table).  For example, if you try to add:   Alfred's Futterkiste,   you will get this prompt:

Which is then followed by run-time error 3075 when you attempt to close the Customers form:

If you click on Debug, you will see that it fails when calling the DLookup function:
Result = DLookup("[CompanyName]", "Customers", _

             "[CompanyName]='" & NewData & "'")
The reason for this failure is that the DLookup function uses apostrophes around the criteria for strings.  The apostrophe in the NewData value interferes with the required syntax, which is shown below:

    For strings: (note the apostrophe before and after the value)

        DLookup("FieldName" , "TableName" , "Criteria= 'string'")

For more information on using DLookup, see: http://www.mvps.org/access/general/gen0018.htm
Tom's Method: Another method that uses a Form to Add a New Record
This method does not use the DLookup function, so one can add values that include an apostrophe.  It does not check for a duplicate entry, which is what the DLookup function above prevents.  However, if the entry is truly a duplicate, then it should be found in the list (and the not in list event procedure won't fire).
The frmCustomers form includes a load event procedure, along with OK and Cancel button event procedures.  If one clicks on OK, we save the record and hide the form (error handling omitted for brevity in the first three procedures):

Code in frmCustomers

Private Sub Form_Load()

   If Not IsNull(Me.OpenArgs) Then

      ' If form's OpenArgs property has a value, assign the contents of OpenArgs to the
      ' CompanyName field. OpenArgs will contain a company name if this form is
      ' opened using the OpenForm method with an OpenArgs argument, as done in 

      ' the Orders form's CustomerID_NotInList event procedure.

      Me![CompanyName] = Me.OpenArgs

      Me.NavigationButtons = False

      Me.cmdOK.Visible = True

      Me.cmdCancel.Visible = True

   End If

End Sub
Private Sub cmdOK_Enter()

      DoCmd.RunCommand acCmdSaveRecord

      Me.Visible = False

End Sub

If the user clicks on Cancel, we undo the change and close the form:

Private Sub cmdCancel_Enter()

      Me.Undo

      DoCmd.Close

End Sub

Code in Orders form for combo box: cboTomsMethod
Private Sub cboTomsMethod_NotInList(NewData As String, Response As Integer)

On Error GoTo ProcError

Dim strResponse As String

strResponse = MsgBox("Customer not found" & vbCrLf & _

  "Do you wish to add this company?", vbYesNo + vbInformation, "Please Respond")

If strResponse = vbYes Then

    DoCmd.OpenForm "frmCustomers", Datamode:=acFormAdd, _

  


 WindowMode:=acDialog, OpenArgs:=NewData

        If IsLoaded2("frmCustomers") Then   'After record is added, we close the hidden form

        
    Response = acDataErrAdded

                DoCmd.Close acForm, "frmCustomers"

        Else

                Response = acDataErrContinue

        End If

 Else

    Response = acDataErrContinue  ' Suppress the standard error message if user




          ' clicks on cancel button on frmCustomers

 End If

ExitProc:

  Exit Sub

ProcError:

  MsgBox "Error " & Err.Number & ": " & Err.Description

  Resume ExitProc

End Sub
The procedure shown above makes a call to the IsLoaded2 function.  (The "2" suffix was added to avoid a name conflict with an existing IsLoaded function in the Northwind database).  Paste the following code into a new module:

Code to add to a new module
Function IsLoaded2(ByVal strFormName As String) As Integer

On Error GoTo ProcError

    IsLoaded2 = False

    Dim frm As Form

    For Each frm In Forms

      If frm.Name = strFormName Then

         IsLoaded2 = True

      End If

    Next frm

ExitProc:

  Exit Function

ProcError:

  MsgBox "Error " & Err.Number & ": " & Err.Description

  Resume ExitProc

End Function

